

第六章

選擇權的敏感性 與美式選擇權的定價

第一節 敏感性的定義

第二節 封閉解程式碼

第三節 差分與微分

第四節 美式選擇權定價公式

選擇權的敏感性在數學上就是偏微分的概念，它描述在其它變數固定下，單一變數的變動對選擇權價格的影響。這可以提供我們對選擇權價格變動的一個概估的工具。本章也介紹一些美式選擇權的封閉公式解，雖然在理論的發展上它們有其歷史意義，然而在實務上，樹狀模型才是美式選擇權的定價的主要方法。

第一節 敏感性的定義

根據 BSM 模型，選擇權的價格可以看成是下面的函數關係，

$$V = f(S, K, T, \sigma, r, y) \dots\dots\dots(6.1.1)$$

其中 S, K, T, σ, r, y 是價格函數的自變數， V 則是價格函數的因變數。所謂敏感性便是指相對於自變數的微小變動下，相對應的因變數變動量。由於執行價格本身是不會變動的，因此根據此一定義，選擇權的價格函數中，我們便有五個不同的敏感性。然而，由於 Delta 的重要性，Delta 本身的敏感性也成為我們研究的對象。下面分項逐一說明，

一、Delta

$$\Delta = \frac{\partial V}{\partial S} \dots\dots\dots(6.1.2)$$

(6.1.2)式的數值稱之為 Delta，又稱為避險比率。因為當投資人放空一單位買權時，需要買入避險的現貨數量便是 Delta。Delta 告訴我們若標的資產價格變動一單位時，相對應的選擇權價格變動量。需注意的是，(6.1.2)的關係式是微分式，因此只適用於很小的標的資產價格變動量。如果標的資產價格變動量很大，(6.1.2)的關係便不會準確。

$$\Delta = \begin{cases} N(d_1), & \text{Call} \\ N(d_1) - 1, & \text{Put} \end{cases} \dots\dots\dots(6.1.3)$$

由於 $N(d_1)$ 恆大於零，因此買權的 Delta 一定大於零。它代表股票價格上漲時，買權上漲的數量。由於買權的報酬為股價超過執行價格的部份，因此股價愈高，買權也愈高，故 Delta 必定為正。事實上，Delta 值就是選擇權價格對股票價格所作的圖形的斜率。對買權而言，當 $S \ll K$ （股票價格遠小於執行價格）時，價格線幾乎是水平的，即斜率為零；當 $S \gg K$ （股票價格遠大於執行價格）時，價格線幾乎是 45 度直線的，即斜率為一；在當 $S = K$ 時，斜率概估為 0.5。相對應的賣權的 Delta 一定小於零，這反應股價愈高，賣權的價值愈低。

二、Gamma

$$\Gamma = \frac{\partial \Delta}{\partial S} = \frac{\partial^2 V}{\partial S^2} \dots\dots\dots(6.1.4)$$

選擇權的 Gamma，衡量股價變動下 Delta 的相對改變情形。由於投資人可由 Delta 值來決定避險的數量，但是 Delta 本身是會改變的，因此 Delta 的變動程度會影響到投資人調整避險部位的頻率。對於一個已執行好的避險部位，如果 Gamma 值很大，一但股價有些微的變動，Delta 值便有大幅的變化，因此避險部位便需立刻去調整。不論買權或賣權，其 Gamma 值都是一樣的，

$$\Gamma = \frac{N'(d_1)}{S\sigma\sqrt{T}} \dots\dots\dots(6.1.5)$$

考慮一極端價外， $S \ll K$ ，的買權，其 Delta 大約為 0。如果股價微量變動，其 Delta 依舊為 0。因此，對一極端價外的買權，Gamma 大約為 0。相同地，處於極端價內， $S \gg K$ ，的買權 Delta 為 1，且如果 S 改變很小的話，Delta 幾乎沒有改變。因此，對極端價內的買權 Gamma 也是大約為 0。當選擇權處於價平時 Gamma 最大。此時只要股票價格稍有變化，買權的 Delta 值有最大的變化。

根據微積分理論，二次微分與選擇權價格對股票價格所作的圖形的曲度有

關。曲度愈彎，Gamma 值愈大，而直線的曲度為零，Gamma 值也為零。對買權與賣權而言，當 $S = K$ 時，圖形曲度最大，因此 Gamma 值最大。當 $S \ll K$ 時，價格線幾乎是水平的，Gamma 為零； $S \gg K$ 時，價格線幾乎是 45 度直線的，Gamma 亦為零。

三、Vega

$$Vega = \frac{\partial V}{\partial \sigma} = S\sqrt{T}N'(d_1) \dots\dots\dots(6.1.6)$$

選擇權的 Vega，衡量選擇權價格相對股價波動性變動的改變情形。雖然在 BSM 模型中假設波動性是固定的，但是現實世界上，因為新的訊息一直產生，波動性的預期也隨時在改變。對買權而言，因為波動性愈大，未來股價上漲的機會也增加，但損失卻不變。對賣權而言，因為波動性愈大，未來股價下跌的機會也增加，但損失卻不變。故 Vega 一定大於零。

由於波動性大都以百分比的方式報價，因此通常 Vega 是衡量 1% 波動性的增加下，選擇權價格的變動量。

四、Theta

$$\Theta = -\frac{\partial V}{\partial T} \dots\dots\dots(6.1.7)$$

Theta 為選擇權價格對到期時間的偏微分的負數，代表隨契約到期時間的減少，對選擇權價值的影響。

$$\Theta = \begin{cases} -S\sigma \frac{N'(d_1)}{2\sqrt{T}} - rKe^{-rT}N(d_2), & Call \\ -S\sigma \frac{N'(d_1)}{2\sqrt{T}} + rKe^{-rT}N(-d_2), & Put \end{cases} \dots\dots\dots(6.1.8)$$

對買權而言，Theta 恆小於零。但是對於賣權而言，Theta 可能為正數。

五、Rho of r

$$Rho_r = \frac{\partial V}{\partial r} \dots\dots\dots(6.1.9)$$

Rho_r 為選擇權價格對利率的偏微分，代表利率的增加，對選擇權價值的影響。

$$Rho = \begin{cases} KTe^{-rT} N(d_2), & Call \\ -KTe^{-rT} N(-d_2), & Put \end{cases} \dots\dots\dots(6.1.10)$$

(6.1.10)式指出買權和利率呈正向關係，隨利率上升買權價格也跟著上漲。賣權則剛好呈現反向關係。由於利率大都以百分比的方式報價，因此通常 Rho_r 是衡量 1%利率的增加下，選擇權價格的變動量。

六、Rho of y

$$Rho_y = \frac{\partial V}{\partial r} \dots\dots\dots(6.1.11)$$

Rho_y 為選擇權價格對資產收益率的偏微分，代表資產收益率的增加，對選擇權價值的影響。

$$Rho_y = \begin{cases} -STe^{-yT} N(d_1), & Call \\ STe^{-yT} N(-d_1), & Put \end{cases} \dots\dots\dots(6.1.12)$$

(6.1.12)式指出買權和呈反向關係。賣權則剛好成現正向關係。由於資產收益率大都以百分比的方式報價，因此通常 Rho_y 是衡量 1%資產收益率的增加下，選擇權價格的變動量。

第二節 封閉解程式碼

前一節的敏感性公式，為不發放股利的 BSM 模型公式，針對一般化的 BSM 模型公式，列示於下，注意我們有兩個 Rho 值，一個是對融資成本利率，一個是對標的資產收益率。

一、Delta

$$\Delta = \begin{cases} e^{-yT} N(d_1), & \text{Call} \\ e^{-yT} [N(d_1) - 1], & \text{Put} \end{cases} \dots\dots\dots(6.2.1)$$

■ 程式碼

```
#01 Public Function GBSDelta(OpClass As String, _
#02     S As Double, K As Double, T As Double, _
#03     sig As Double, r As Double, y As Double) _
#04     As Double
#05
#06     Dim d1 As Double
#07
#08     d1 = (Log(S/K)+(r-y+sig^2/2)*T)/(sig*Sqr(T))
#09
#10     If OpClass = "C" Then
#11         GBSDelta = Exp(-y * T) * NorCdf(d1)
#12     ElseIf OpClass = "P" Then
#13         GBSDelta = Exp(-y * T) * (NorCdf(d1)-1)
#14     End If
#15
#16 End Function
```

程式 6.2.1(Book1.xls/Module1)

二、Gamma

$$\Gamma = \frac{e^{-yT} N'(d_1)}{S\sigma\sqrt{T}} \dots\dots\dots(6.2.2)$$

■ 程式碼

```
#01 Public Function GBSGamma(S As Double, K As Double, _
#02     T As Double, sig As Double, r As Double, _
#03     y As Double) As Double
#04
#05     Dim d1 As Double
#06
#07     d1 = (Log(S/K)+(r-y+sig^2/2)*T) / (sig*Sqr(T))
#08     GBSGamma = Exp(-y*T)*NorPdf(d1) / (S*sig*Sqr(T))
#09
#10 End Function
```

程式 6.2.2(Book1.xls/Module1)

三、Vega

$$Vega = Se^{-yT} \sqrt{T} N'(d_1) \dots\dots\dots(6.2.3)$$

■ 程式碼

```
#01 Public Function GBSVega(S As Double, _
#02     K As Double, T As Double, sig As Double, _
#03     r As Double, y As Double) As Double
#04
#05     Dim d1 As Double
#06
#07     d1 = (Log(S/K)+(r-y+sig^2/2)*T) / (sig*Sqr(T))
#08     GBSVega = S*Exp(-y*T)*NorPdf(d1)*Sqr(T)
#09
#10 End Function
```

程式 6.2.3(Book1.xls/Module1)

四、Theta

$$\Theta = \begin{cases} -Se^{-yT} \sigma \frac{N'(d_1)}{2\sqrt{T}} + ySe^{-yT} N(d_1) - rKe^{-rT} N(d_2), & \text{Call} \\ -Se^{-yT} \sigma \frac{N'(d_1)}{2\sqrt{T}} - ySe^{-yT} N(-d_1) + rKe^{-rT} N(-d_2), & \text{Put} \end{cases}$$

.....(6.2.4)

■ 程式碼

```
#01 Public Function GBSTheta(OpClass As String, _
#02     S As Double, K As Double, T As Double, _
#03     sig As Double, r As Double, y As Double) _
#04     As Double
#05
#06     Dim d1 As Double, d2 As Double
#07
#08     d1 = (Log(S/K) + (r-y+sig^2/2)*T) / (sig*Sqr(T))
#09     d2 = d1 - sig * Sqr(T)
#10
#11     If OpClass = "C" Then
#12         GBSTheta = -S*Exp(-y*T)*NorPdf(d1)*sig _
#13             /(2*Sqr(T))+y*S*Exp(-y*T)*NorCdf(d1) _
#14             -r*K*Exp(-r*T)*NorCdf(d2)
#15     ElseIf OpClass = "P" Then
#16         GBSTheta = -S*Exp(-y*T)*NorPdf(d1)*sig _
#17             /(2*Sqr(T))-y*S*Exp(-y*T)*NorCdf(-d1) _
#18             +r*K*Exp(-r*T)*NorCdf(-d2)
#19     End If
#20
#21 End Function
```

程式 6.2.4(Book1.xls/Module1)

五、Rho_r

$$Rho = \begin{cases} KTe^{-rT}N(d_2), & \text{Call} \\ -KTe^{-rT}N(-d_2), & \text{Put} \end{cases} \dots\dots\dots(6.2.5)$$

■ 程式碼

```
#01 Public Function GBSRho_r(OpClass As String, _
#02     S As Double, K As Double, T As Double, _
#03     sig As Double, r As Double, y As Double) _
#04     As Double
#05
#06     Dim d1 As Double, d2 As Double
#07
#08     d1 = (Log(S/K) + (r-y+sig^2/2)*T) / (sig*Sqr(T))
#09     d2 = d1 - sig * Sqr(T)
#10
#11     If OpClass = "C" Then
#12         If (r - y) <> 0 Then
#13             GBSRho_r = T*K*Exp(-r*T)*NorCdf(d2)
#14         Else
#15             GBSRho_r= -T* GBSOption(OpClass, S, K, T, _
#16                 sig , r, y)
#17         End If
#18     ElseIf OpClass = "P" Then
#19         If (r - y) <> 0 Then
#20             GBSRho_r = -T*K*Exp(-r*T)*NorCdf(-d2)
#21         Else
#22             GBSRho_r = -T * GBSOption(OpClass, S, K, _
#23                 T, sig, r, y)
#24         End If
#25     End If
#26
#27 End Function
```

程式 6.2.5(Book1.xls/Module1)

六、Rho_y

$$Rho = \begin{cases} -STe^{-yT}N(d_1), & \text{Call} \\ STe^{-yT}N(-d_1), & \text{Put} \end{cases} \dots\dots\dots(6.2.6)$$

■ 程式碼

```
#01 Public Function GBSRho_y(OpClass As String, _
#02     S As Double, K As Double, T As Double, _
#03     sig As Double, r As Double, y As Double) _
#04     As Double
#05
#06     Dim d1 As Double
#07
#08     d1 = (Log(S/K) + (r-y+sig^2/2)*T)/(sig*Sqr(T))
#09
#10     If OpClass = "C" Then
#11         GBSRho_y = -T*S*Exp(-y*T) * NorCdf(d1)
#12     ElseIf OpClass = "P" Then
#13         GBSRho_y = T*S*Exp(-y*T) * NorCdf(-d1)
#14     End If
#15
#16 End Function
```

程式 6.2.6(Book1.xls/Module1)

■ 電腦程式範例

考慮下面的資料， $S = 100$ 、 $K = 100$ 、 $T = 1$ 年、 $\sigma = 30\%$ 、 $r = 8\%$ 、 $y = 2\%$ ，利用(6.2.1)

至(6.2.6)式，我們可求得買權與賣權比較靜態的各個數值，

$$d_1 = \frac{\ln(100/100) + (8\% - 2\% + 0.5 \times 0.3^2) \times 1}{0.3\sqrt{1}} = 0.3500$$

$$d_1 = \frac{\ln(100/100) - (8\% - 2\% + 0.5 \times 0.3^2) \times 1}{0.3\sqrt{1}} = 0.0500$$

```

#01 Public Sub Op_Calculate()
#02   Dim S As Double
#03   Dim K As Double
#04   Dim T As Double
#05   Dim sig As Double
#06   Dim r As Double
#07   Dim y As Double
#08   Dim C_Value As Double
#09   Dim P_Value As Double
#10   Dim d1 As Double, d2 As Double
#11
#12   S = Worksheets("Sheet1").Cells(2, 1).Value
#13   K = Worksheets("Sheet1").Cells(2, 2).Value
#14   T = Worksheets("Sheet1").Cells(2, 3).Value
#15   sig = Worksheets("Sheet1").Cells(2, 4).Value
#16   r = Worksheets("Sheet1").Cells(2, 5).Value
#17   y = Worksheets("Sheet1").Cells(2, 6).Value
#18
#19   d1 = (Log(S/K) + (r-y+sig^2/2)*T) / (sig*Sqr(T))
#20   d2 = d1 - sig * Sqr(T)
#21   Worksheets("Sheet1").Cells(5, 5).Value = d1
#22   Worksheets("Sheet1").Cells(5, 6).Value = d2
#23   Worksheets("Sheet1").Cells(6, 5).Value = NorPdf(d1)
#24   Worksheets("Sheet1").Cells(6, 6).Value = NorPdf(d2)
#25   Worksheets("Sheet1").Cells(7, 5).Value = NorCdf(d1)
#26   Worksheets("Sheet1").Cells(7, 6).Value = NorCdf(d2)
#27
#28   Worksheets("Sheet1").Cells(9, 5).Value = -d1
#29   Worksheets("Sheet1").Cells(9, 6).Value = -d2
#30   Worksheets("Sheet1").Cells(10, 5).Value=NorPdf(-d1)
#31   Worksheets("Sheet1").Cells(10, 6).Value=NorPdf(-d2)
#32   Worksheets("Sheet1").Cells(11, 5).Value=NorCdf(-d1)
#33   Worksheets("Sheet1").Cells(11, 6).Value=NorCdf(-d2)
#34
#35   C_Value = GBSOption("C", S, K, T, sig, r, y)
#36   P_Value = GBSOption("P", S, K, T, sig, r, y)
#37   Worksheets("Sheet1").Cells(5, 2).Value = C_Value
#38   Worksheets("Sheet1").Cells(5, 3).Value = P_Value
#39
#40   C_Value = GBSDelta("C", S, K, T, sig, r, y)
#41   P_Value = GBSDelta("P", S, K, T, sig, r, y)
#42   Worksheets("Sheet1").Cells(6, 2).Value = C_Value
#43   Worksheets("Sheet1").Cells(6, 3).Value = P_Value
#44
#45   C_Value = GBSGamma(S, K, T, sig, r, y)
#46   P_Value = GBSGamma(S, K, T, sig, r, y)
#47   Worksheets("Sheet1").Cells(7, 2).Value = C_Value
#48   Worksheets("Sheet1").Cells(7, 3).Value = P_Value

```

```
#49
#50 C_Value = GBSVega(S, K, T, sig, r, y)
#51 P_Value = GBSVega(S, K, T, sig, r, y)
#52 Worksheets("Sheet1").Cells(8, 2).Value = C_Value
#53 Worksheets("Sheet1").Cells(8, 3).Value = P_Value
#54
#55 C_Value = GBSTheta("C", S, K, T, sig, r, y)
#56 P_Value = GBSTheta("P", S, K, T, sig, r, y)
#57 Worksheets("Sheet1").Cells(9, 2).Value = C_Value
#58 Worksheets("Sheet1").Cells(9, 3).Value = P_Value
#59
#60 C_Value = GBSRho_r("C", S, K, T, sig, r, y)
#61 P_Value = GBSRho_r("P", S, K, T, sig, r, y)
#62 Worksheets("Sheet1").Cells(10, 2).Value = C_Value
#63 Worksheets("Sheet1").Cells(10, 3).Value = P_Value
#64
#65 C_Value = GBSRho_y("C", S, K, T, sig, r, y)
#66 P_Value = GBSRho_y("P", S, K, T, sig, r, y)
#67 Worksheets("Sheet1").Cells(11, 2).Value = C_Value
#68 Worksheets("Sheet1").Cells(11, 3).Value = P_Value
#69
```

```
#70 End Sub
```

程式 6.2.7(Book1.xls/Sheet1)

由於 $\text{Call Delta} = 0.6242$ ，因此資產價格上漲\$1元，買權上漲 $\$1 \times 0.6242 = \0.6242 。隨到期日接近一日，買權下跌 $1/365 \times -8.1085 = -\0.022215 。利率上漲1%，買權上漲 $0.01 \times 47.9971 = \$0.479971$ 。波動性上漲1%，買權上漲 $0.01 \times 36.7810 = \$0.367810$ 。資產收益率上漲1%，買權上漲 $0.01 \times -62.4215 = -\0.624215 。對賣權一樣可以得到類似的推論結果，請讀者自行分析。

	A	B	C	D	E	F	G	H
1	S	K	T	sig	r	y		
2	100	100	1	30%	8%	2%		
3								
4		Call	Put		d1	d2		
5	Premium	14.4244	8.7161	Value	0.35000	0.05000		
6	Delta	0.6242	-0.3560	NorPdf	0.37524	0.39844		
7	Gamma	0.0123	0.0123	NorCdf	0.63683	0.51995		
8	Vega	36.7810	36.7810		-d1	-d2		
9	Theta	-8.1085	-2.6840	Value	-0.3500	-0.0500		
10	Rho_r	47.9971	-44.3145	NorPdf	0.37524	0.39844		
11	Rho_y	-62.4215	35.5984	NorCdf	0.36317	0.48005		
12								
13								
14								
15								
16								
17								

圖 6.2.1(Book1.xls/Sheet1)

第三節 差分與微分

前述(6.2.1)到(6.2.6)的敏感性公式，乃是針對一般化的 Black-Scholes 公式，利用微分而得到的封閉解公式。由於敏感性數值不論對交易人員或風管人員都有很重要的意義，因此不論對於何種選擇權，我們應該都要算出這些數值。然而在實務的運作上，有兩個議題是我們必須面對的。

一、敏感性封閉解的實作問題

首先，並不是每一種選擇權都像陽春型的選擇權，可以導出 Black-Scholes 公式般的封閉解定價公式。即使可以有封閉解，要進一步再導出各個敏感性公式也相當的複雜。在後面有關異種(Exotic)選擇權的章節，我們可以看到許多選擇權的定價公式都是相當複雜的。

其次，(6.2.1)到(6.2.6)的敏感性公式都是利用微分而得到的。所謂微分乃是計算相關的自變數在微小的變動量下，相對應的因變數變動量。現實的世界上，我們所處理的自變數，如股價、利率和波動性，它們的變動量都是離散的(Discrete)，而不是連續變動。因此，利用連續概念求得的微分公式，未必適用於現實的離散變動環境。

爲了要克服前述的兩個問題，我們可以利率差分的方式，直接求得敏感性的數值。以買權 Delta 值爲例，根據定義 Delta 爲股價變動下，選擇權價格的變動量，可表示爲：

$$Delta = \frac{C(S + \Delta S, K, T, \sigma, r, y) - C(S, K, T, \sigma, r, y)}{(S + \Delta S) - S} \dots\dots\dots(6.3.1)$$

(6.3.1)式告訴我們，Delta 的分子是由資產價格變動後的選擇權價格，減去資產價格變動前的選擇權價格；分母則是資產價格的變動量。因此，我們只要計算兩次的選擇權價格，便可求得 Delta 值了。

■ 數值範例

考慮下面的資料， $S = 100$ 、 $K = 100$ 、 $T = 1$ 年、 $\sigma = 30\%$ 、 $r = 8\%$ 、 $y = 2\%$ ，可求得買權價格爲 14.4256。當 $S = 101$ ，其他參數不變下，可求得買權價格爲 15.0559。利用(6.3.1)式，可求得 Delta 爲 0.6303。

$$Delta = \frac{C(S + \Delta S) - C(S)}{(S + \Delta S) - S} = \frac{15.0559 - 14.4256}{101 - 100} = 0.6303。$$

二、差分法的敏感性

由前述可知，敏感性可由差分式比率求得，但是(6.3.1)式的差分式是由標的資產價格加上變化量而得。這樣的差分式稱之爲前向差分(Forward Difference)，這樣求得的差分値並不適合代表在原價格上的差分値。我們可以下面的差分式，來求得原價格上的差分値。

$$\Delta = \frac{\partial V}{\partial S} \approx \frac{V(S + \Delta S) - V(S - \Delta S)}{2\Delta S} \dots\dots\dots(6.3.2)$$

作法是往前與往後等量變動，計算差額，在除以兩倍的變動量。通常我們令變動總量為百分之一的原價格， $2\Delta S = 0.01S$ 。因此， $\Delta S = 0.5\% S$ 。

至於其他的敏感性數值，也可同理求得。下面為這些差分式。

$$Vega = \frac{\partial V}{\partial \sigma} \approx \frac{V(\sigma + \Delta \sigma) - V(\sigma - \Delta \sigma)}{2\Delta \sigma} \dots\dots\dots(6.3.3)$$

$$\Theta = -\frac{\partial V}{\partial T} \approx -\frac{V(T + \Delta T) - V(T - \Delta T)}{2\Delta T} \dots\dots\dots(6.3.4)$$

$$Rho_r = \frac{\partial V}{\partial r} \approx \frac{V(r + \Delta r) - V(r - \Delta r)}{2\Delta r} \dots\dots\dots(6.3.5)$$

$$Rho_y = \frac{\partial V}{\partial y} \approx \frac{V(y + \Delta y) - V(y - \Delta y)}{2\Delta y} \dots\dots\dots(6.3.6)$$

由於 Gamma 為二次微分，因此它的差分式不同於其他的差分式。下面為其公式。

$$\Gamma = \frac{\partial^2 V}{\partial S^2} \approx \frac{V(S + \Delta S) - 2V(S) + V(S - \Delta S)}{\Delta S^2} \dots\dots\dots(6.3.7)$$

■ 電腦程式範例

考慮下面的資料， $S = 100$ 、 $K = 100$ 、 $T = 1$ 年、 $\sigma = 30\%$ 、 $r = 8\%$ 、 $y = 2\%$ ，利用(6.3.2)至(6.3.7)式，我們可求得買權與賣權比較靜態的各個數值，程式(6.3.1)為差分計算的範例程式。




圖 6.3.1(Book2.xls/Sheet1)

```

#01 Public Sub Op_Calculate()
#02     Dim S As Double
#03     Dim K As Double
#04     Dim T As Double
#05     Dim sig As Double
#06     Dim r As Double
#07     Dim y As Double
#08     Dim V As Double, V0 As Double, V1 As Double
#09     Dim delta As Double
#10
#11     S = Worksheets("Sheet1").Cells(2, 1).Value
#12     K = Worksheets("Sheet1").Cells(2, 2).Value
#13     T = Worksheets("Sheet1").Cells(2, 3).Value
#14     sig = Worksheets("Sheet1").Cells(2, 4).Value
#15     r = Worksheets("Sheet1").Cells(2, 5).Value
#16     y = Worksheets("Sheet1").Cells(2, 6).Value
#17
#18     delta = 0.01 * S / 2
#19     V0 = GBSOption("C", S - delta, K, T, sig, r, y)
#20     V1 = GBSOption("C", S + delta, K, T, sig, r, y)
#21     Worksheets("Sheet1").Cells(6, 2).Value = _
#22         (V1 - V0) / (2 * delta)
#23

```



```

#24    V = GBSOption("C", S, K, T, sig, r, y)
#25    Worksheets("Sheet1").Cells(7, 2).Value = _
#26        (V1 - 2 * V + V0) / (delta ^ 2)
#27
#28    delta = 0.01 * sig / 2
#29    V0 = GBSOption("C", S, K, T, sig - delta, r, y)
#30    V1 = GBSOption("C", S, K, T, sig + delta, r, y)
#31    Worksheets("Sheet1").Cells(8, 2).Value = _
#32        (V1 - V0) / (2 * delta)
#33
#34    delta = 0.01 * T / 2
#35    V0 = GBSOption("C", S, K, T - delta, sig, r, y)
#36    V1 = GBSOption("C", S, K, T + delta, sig, r, y)
#37    Worksheets("Sheet1").Cells(9, 2).Value = _
#38        -(V1 - V0) / (2 * delta)
#39
#40    delta = 0.01 * r / 2
#41    V0 = GBSOption("C", S, K, T, sig, r - delta, y)
#42    V1 = GBSOption("C", S, K, T, sig, r + delta, y)
#43    Worksheets("Sheet1").Cells(10, 2).Value = _
#44        (V1 - V0) / (2 * delta)
#45
#46    delta = 0.01 * y / 2
#47    V0 = GBSOption("C", S, K, T, sig, r, y - delta)
#48    V1 = GBSOption("C", S, K, T, sig, r, y + delta)
#49    Worksheets("Sheet1").Cells(11, 2).Value = _
#50        (V1 - V0) / (2 * delta)
#51 End Sub

```

程式 6.3.1(Book2.xls/Sheet1)

第四節 美式選擇權定價公式

由於美式選擇權具有提前執行的權利，因此相對於歐式的到期執行的條件，複雜度增加不少。這是因為提前執行的時機本身就帶有不確定性。再加上資產價格的不確定性，我們便具有雙重的隨機性。很自然的，我們需要雙元的分配去描述此一現象。

儘管學者發表了許多的美式定價公式，但大都僅是近似公式。事實上，針對美式選擇權，實務上乃採用樹狀模型來計算價格。我們將在後面的章節中，介紹二元樹模型下的美式選擇權定價方法。本節中將介紹一些較為著名的模型

。

一、標準雙元常態分配

兩隨機變數 X 、 Y 機率分配稱之為標準雙元常態分配，若其機率密度函數為下式

$$\Psi(x, y, \rho) = \frac{1}{2\pi\sqrt{1-\rho^2}} e^{-\frac{1}{2} \frac{(x^2+y^2-2\rho \cdot x \cdot y)}{(1-\rho^2)}}$$

其累積機率密度函數 $M(a, b, \rho)$ 定義為

$$M(a, b, \rho) = \int_{-\infty}^a \int_{-\infty}^b \Psi(u, v, \rho) \cdot du \cdot dv \dots\dots\dots (6.4.1)$$

使用數值積分估計(6.4.1)式雖然可行，但依舊受限於計算時間，因此並不可行。

利用解析函數的近似方法，為可行的方案，(6.4.1)式的近似式可表示為，

$$a1 = \frac{a}{\sqrt{2(1-\rho^2)}} \quad , \quad b1 = \frac{b}{\sqrt{2(1-\rho^2)}} \quad ,$$

$$A_1 = 0.24840615 \quad , \quad A_2 = 0.39233107 \quad , \quad A_3 = 0.21141819 \quad ,$$

$$A_4 = 0.03324666 \quad , \quad A_5 = 0.00082485334$$

$$B_1 = 0.10024215 \quad , \quad B_2 = 0.48281397 \quad , \quad B_3 = 1.0609498 \quad ,$$

$$B_4 = 1.7797294 \quad , \quad B_5 = 2.6697604$$

■ (i) $a \leq 0, b \leq 0, \rho \leq 0$

$$\Omega = \sum_{i=1}^5 \sum_{j=1}^5 A_i A_j e^{[a1(2-B_i-a1)+b1(2-B_j-b1)+2\rho(B_i-a1)(B_j-b1)]}$$

$$M(a, b, \rho) = \frac{\sqrt{1-\rho^2}}{\pi} \times \Omega$$

- (ii) $a \leq 0, b \geq 0, \rho \geq 0$

$$M(a, b, \rho) = N(a) - M(a, -b, -\rho)$$

- (iii) $a \geq 0, b \leq 0, \rho \geq 0$

$$M(a, b, \rho) = N(b) - M(-a, b, -\rho)$$

- (iv) $a \geq 0, b \geq 0, \rho \leq 0$

$$M(a, b, \rho) = N(a) + N(b) - 1 + M(-a, -b, \rho)$$

- (v) $a \times b \times \rho > 0$

$$M(a, b, \rho) = M(a, 0, \rho_1) + M(b, 0, \rho_2) - \Theta$$

$$\rho_1 = \frac{(\rho \cdot a - b) \times \text{Sgn}(a)}{\sqrt{a^2 - 2\rho ab + b^2}},$$

$$\rho_2 = \frac{(\rho \cdot b - a) \times \text{Sgn}(b)}{\sqrt{a^2 - 2\rho ab + b^2}},$$

$$\Theta = \frac{1 - \text{Sgn}(a) \times \text{Sgn}(b)}{4},$$

$$\text{Sgn}(x) = \begin{cases} 1, & x > 0 \\ 0, & x = 0 \\ -1, & x < 0 \end{cases}.$$

程式 6.4.1 即為此解析函數近似方法的計算函數，此函數需要輸入三個參數，對應到(6.4.1)式中 $M(a, b, \rho)$ 的參數。至於呼叫此函數的範例，留給讀者自行練習。

```
#01 Public Function BiNorCdf(a As Double, b As Double, _
#02     rho As Double) As Double
#03
#04     Dim X As Variant, y As Variant
#05     Dim rho1 As Double, rho2 As Double, delta As Double
#06     Dim a1 As Double, b1 As Double, Sum As Double
#07     Dim I As Integer, j As Integer
#08
#09     X = Array(0.24840615, 0.39233107, 0.21141819, _
#10         0.03324666, 0.00082485334)
#11     y = Array(0.10024215, 0.48281397, 1.0609498, _
#12         1.7797294, 2.6697604)
#13     a1 = a / Sqr(2 * (1 - rho ^ 2))
#14     b1 = b / Sqr(2 * (1 - rho ^ 2))
#15
#16     If a <= 0 And b <= 0 And rho <= 0 Then
#17         Sum = 0
#18         For I = 1 To 5
#19             For j = 1 To 5
#20                 Sum = Sum+X(I)*X(j)*Exp(a1*(2*y(I)-a1) _
#21                     +b1*(2*y(j)-b1)+2*rho*(y(I)-a1) _
#22                     *(y(j)-b1))
#23             Next
#24         Next
#25         BiNorCdf = Sqr(1 - rho ^ 2) / Pi * Sum
#26     ElseIf a <= 0 And b >= 0 And rho >= 0 Then
#27         BiNorCdf = NorCdf(a) - BiNorCdf(a, -b, -rho)
#28     ElseIf a >= 0 And b <= 0 And rho >= 0 Then
#29         BiNorCdf = NorCdf(b) - BiNorCdf(-a, b, -rho)
#30     ElseIf a >= 0 And b >= 0 And rho <= 0 Then
#31         BiNorCdf = NorCdf(a) + NorCdf(b) - 1 + _
#32             BiNorCdf(-a, -b, rho)
#33     ElseIf a * b * rho > 0 Then
#34         rho1 = (rho*a-b)*Sgn(a)/Sqr(a^2-2*rho*a*b+b^2)
#35         rho2 = (rho*b-a)*Sgn(b)/Sqr(a^2-2*rho*a*b+b^2)
#36         delta = (1 - Sgn(a) * Sgn(b)) / 4
#37         BiNorCdf = BiNorCdf(a, 0, rho1) + _
#38             BiNorCdf(b, 0, rho2) - delta
#39     End If
#40 End Function
```

程式 6.4.1(Book3.xls/Module1)

二、間斷股利的買權

Roll, R.(1977)，Geske, R.(1979)和 Whaley, R.(1981)發展出了間斷股利的美式買權價格公式。若股利為 D_t ，則公式為

$$\begin{aligned}
 C = & (S - D_t e^{-rt})N(b_1) + (S - D_t e^{-rt})M(a_1, -b_1; -\sqrt{t/T}) \\
 & - Ke^{-rT}M(a_2, -b_2; -\sqrt{t/T}) \\
 & - (K - D_t)e^{-rT}N(b_2) \dots\dots\dots(6.4.2)
 \end{aligned}$$

其中

$$a_1 = \frac{\ln\left[\frac{(S - D_t e^{-rT})}{K}\right] + (r + \sigma^2/2)T}{\sigma\sqrt{T}}$$

$$a_2 = a_1 - \sigma\sqrt{t}$$

$$b_1 = \frac{\ln\left[\frac{(S - D_t e^{-rT})}{\bar{S}}\right] + (r + \sigma^2/2)T}{\sigma\sqrt{T}}$$

$$d_2 = d_1 - \sigma\sqrt{t}$$

函數 $M(a, b, \rho)$ 代表第一個參數小於 a ，第二個參數小於 b ，相關係數為 ρ 的標準雙元常態分配(Standardized Bivariate Normal Distribution)之累積機率密度函數。變數 \bar{S} 為下面方程式的解：

$$C^E(\bar{S}, t) = \bar{S} + D_t - K$$

其中 $C^E(\bar{S}, t)$ 為標的資產價格 \bar{S} ，到期日 t 的歐式買權價格。換句話說， \bar{S} 代表關鍵資產價格，當價格高於 \bar{S} 時提前執行為最佳。如果提前執行不是最佳，則 $\bar{S} = \infty$ ， $d_1 = \infty$ ， $d_2 = \infty$ 且(6.4.2)式退化成(4.3.2)。

```

#01 Public Function RollGeskeWhaley(S As Double, _
#02     K As Double, t1 As Double, T2 As Double, _
#03     r As Double, D As Double, sig As Double) As Double
#04     't1 time to dividend payout
#05     'T2 time to option expiration
#06
#07     Dim Sx As Double, I As Double
#08     Dim a1 As Double, a2 As Double
#09     Dim b1 As Double, b2 As Double
#10     Dim HighS As Double, LowS As Double
#11     Dim epsilon As Double
#12     Dim ci As Double, infinity As Double
#13
#14     infinity = 1000000000
#15     epsilon = 0.00001
#16     Sx = S - D * Exp(-r * t1)
#17
#18     If D <= K * (1 - Exp(-r * (T2 - t1))) Then
#19         RollGeskeWhaley = BSOOption("C", Sx, K, T2, sig, r)
#20         Exit Function
#21     End If
#22
#23     ci = BSOOption("C", S, K, T2 - t1, sig, r)
#24     HighS = S
#25
#26     While (ci - HighS - D + K) > 0 And HighS < infinity
#27         HighS = HighS * 2
#28         ci = BSOOption("C", HighS, K, T2 - t1, sig, r)
#29     Wend
#30
#31     If HighS > infinity Then
#32         RollGeskeWhaley = BSOOption("C", Sx, K, T2, sig, r)
#33         Exit Function
#34     End If
#35
#36     LowS = 0
#37     I = HighS * 0.5
#38     ci = BSOOption("C", I, K, T2 - t1, sig, r)
#39

```

```

#40 While ((Abs(ci - I - D + K) > epsilon) And _
#41 (HighS - LowS > epsilon))
#42     If (ci - I - D + K) < 0 Then
#43         HighS = I
#44     Else
#45         LowS = I
#46     End If
#47     I = (HighS + LowS) / 2
#48     ci = BSOOption("C", I, K, T2 - t1, sig, r)
#49 Wend
#50
#51 a1 = (Log(Sx/K) + (r+sig^2/2)*T2) / (sig*Sqr(T2))
#52 a2 = a1 - sig * Sqr(T2)
#53 b1 = (Log(Sx/I) + (r+sig^2/2)*t1) / (sig*Sqr(t1))
#54 b2 = b1 - sig * Sqr(t1)
#55
#56 RollGeskeWhaley = Sx * NorCdf(b1) _
#57     + Sx * BiNorCdf(a1,-b1,-Sqr(t1/T2)) _
#58     - K * Exp(-r*T2)*BiNorCdf(a2,-b2,-Sqr(t1/T2)) _
#59     - (K-D) * Exp(-r*t1)*NorCdf(b2)
#60 End Function
程式 6.4.2(Book3.xls/Module1)

```

在程式 6.4.2 中，我們用 $t1$ 代表股利發放的時間，以 $T2$ 代表選擇權到期日。這是因為在 VBA 中，變數名稱用大、小寫表示，是視為相同的。

三、連續股利的情況

由於並沒有正確的程序來評價連續股利的美式買權，因此只能使用數值方法去求得近似解。Barone-Adesi, G.和 Whaley, R.(1987)提出下面方法。

$$C^A = C^E(S, K, T) + A_2 \left(\frac{S}{\bar{S}} \right)^{q_2}, \quad S < \bar{S} \quad \dots\dots\dots(6.4.3)$$

$$C^A = S - K, \quad S \geq \bar{S}$$

其中

$$A_2 = \left(\frac{\bar{S}}{q_2} \right) \{1 - e^{-(r-y)T} N[d_1(\bar{S})]\}$$

$$d_1(\bar{S}) = \frac{\ln(\bar{S}/K) + (y + \sigma^2/2)T}{\sigma\sqrt{T}}$$

$$q_2 = \frac{-(N-1) + \sqrt{(N-1)^2 + 4M/K}}{2}$$

$$M = \frac{2r}{\sigma^2}, \quad N = \frac{2y}{\sigma^2}, \quad K = (1 - e^{-rT})$$

C^E = 相對應的歐式權利金，(6.4.3)式中 \bar{S} 表提前執行的關鍵價格。若目前價格高於 \bar{S} ，則買權應立刻執行。 \bar{S} 可利用遞迴方式由下式求出

$$\bar{S} - K = C^E(\bar{S}) + \{1 - e^{-(r-y)T} N[d_1(\bar{S})]\} \bar{S} / q_2$$

換句話說，(6.4.3)式指出如果 $S < \bar{S}$ ，則美式買權等於相對應的歐式買權加上提前執行的權利金 $A_2(S/\bar{S})^{q_2}$ 。若 $S \geq \bar{S}$ ，美式買權價格即為其內含價值。


```

#01 '// Barone-Adesi and Whaley (1987) American
#02 Public Function BAWAmericanApprox(CallPutFlag As
#03 string, S As Double, K As Double, T As Double, r As Double,
#04 y As Double, sig As Double) As Double
#05
#06     If CallPutFlag = "c" Then
#07         BAWAmericanApprox = BAWAmericanCallApprox(S, K,
#08             T, r, y, sig)
#09     ElseIf CallPutFlag = "p" Then
#10         BAWAmericanApprox = BAWAmericanPutApprox(S, K,
#11             T, r, y, sig)
#12     End If
#13
#14 End Function
#15
#16
#17 '// American call
#18 Private Function BAWAmericanCallApprox(S As Double,
#19 K As Double, T As Double, r As Double, y As Double,
#20 sig As Double) As Double
#21
#22     Dim Sk As Double, n As Double, CP As Double
#23     Dim d1 As Double, Q2 As Double, a2 As Double
#24
#25     If (r - y) >= r Then
#26         BAWAmericanCallApprox = GBSOption("c", S, K,
#27             T, sig, r, y)
#28     Else
#29         Sk = Kc(K, T, r, y, sig)
#30         n = 2 * (r - y) / sig ^ 2
#31         CP = 2 * r / (sig ^ 2 * (1 - Exp(-r * T)))
#32         d1 = (Log(Sk / K) + (r - y + sig ^ 2 / 2) * T)
#33             / (sig * Sqr(T))
#34         Q2 = (-(n - 1) + Sqr((n - 1) ^ 2 + 4 * CP)) / 2
#35         a2 = (Sk / Q2) * (1 - Exp(-y * T) * NorCdf(d1))
#36         If S < Sk Then
#37             BAWAmericanCallApprox = GBSOption("c", S, K,
#38                 T, sig, r, y) + a2 * (S / Sk) ^ Q2
#39         Else
#40             BAWAmericanCallApprox = S - K
#41         End If
#42     End If
#43 End Function
#44
#45

```

```

#46 '// Newton Raphson to solve critical price for a Call
#47 Private Function Kc(K As Double, T As Double, r As Double,
#48 y As Double, sig As Double) As Double
#49
#50     Dim n As Double, m As Double
#51     Dim Su As Double, Si As Double
#52     Dim h2 As Double, A As Double
#53     Dim d1 As Double, Q2 As Double, q2u As Double
#54     Dim LHS As Double, RHS As Double
#55     Dim bi As Double, E As Double
#56
#57     '// Calculation of seed value, Si
#58     n = 2 * (r - y) / sig ^ 2
#59     m = 2 * r / sig ^ 2
#60     q2u = (-(n - 1) + Sqr((n - 1) ^ 2 + 4 * m)) / 2
#61     Su = K / (1 - 1 / q2u)
#62     h2 = -((r-y)*T + 2*sig*Sqr(T)) * K / (Su - K)
#63     Si = K + (Su - K) * (1 - Exp(h2))
#64
#65     A = 2 * r / (sig ^ 2 * (1 - Exp(-r * T)))
#66     d1 = (Log(Si/K) + (r-y+sig^2/2)*T) / (sig * Sqr(T))
#67     Q2 = (-(n - 1) + Sqr((n - 1) ^ 2 + 4 * A)) / 2
#68     LHS = Si - K
#69     RHS = GBSOption("c",Si,K,T,sig,r,y) + (1-Exp(-y*T)
#70         * NorCdf(d1)) * Si / Q2
#71     bi = Exp(-y*T) * NorCdf(d1) * (1-1/Q2) + (1-Exp(-y*T)
#72         * NorCdf(d1) / (sig * Sqr(T))) / Q2
#73     E = 0.000001
#74
#75     '// Newton Raphson to find critical price Si
#76     While Abs(LHS - RHS) / K > E
#77         Si = (K + RHS - bi * Si) / (1 - bi)
#78         d1 = (Log(Si/K)+(r-y+sig^2/2)*T)/(sig*Sqr(T))
#79         LHS = Si - K
#80         RHS = GBSOption("c",Si,K,T,sig,r,y)
#81             +(1-Exp(-y*T)* NorCdf(d1)) * Si / Q2
#82         bi = Exp(-y* T)*CND(d1)*(1-1/Q2) + (1-Exp(-y*T)
#83             * NorPdf(d1) / (sig * Sqr(T))) / Q2
#84     Wend
#85
#86     Kc = Si
#87 End Function

```

程式 6.4.3(Book3.xls/Module1)

程式 6.4.3 即為 Barone-Adesi 和 Whaley(1987)的演算程式碼，#01~#14 為包裝的程式碼，根據買權或賣權分別呼叫相關的函數。#17~#43 為(6.4.3)式的買權程式碼，其中提前執行的關鍵價格 \bar{S} ，可由#46~#87 利用 Newton-Raphson 法求得。

四、連續與不發放股利的賣權

此法也是由 Barone-Adesi 和 Whaley 提出，對發放連續股利的美式賣權，公式為

$$P^A = P^E(S) + A_1 \left(\frac{S}{\bar{S}} \right)^{q_1}, \quad S > \bar{S} \quad \text{.....(6.4.4)}$$

$$P^A = K - S, \quad S \leq \bar{S}$$

其中

$$A_1 = -\left(\frac{\bar{S}}{q_1} \right) \{1 - e^{-(r-y)T} N[-d_1(\bar{S})]\}$$

$$q_1 = \frac{-(N-1) - \sqrt{(N-1)^2 + 4M/K}}{2}$$

$$M = \frac{2r}{\sigma^2}, \quad N = \frac{2y}{\sigma^2}, \quad K = (1 - e^{-rT})$$

d_1 則與(6.4.3)式中的 d_1 相同。 P^E = 相對應的歐式權利金， \bar{S} 代表若目前價格低於它，賣權便須立刻執行的關鍵價格(critical value)。 \bar{S} 可以遞迴方式由下式求出

$$K - \bar{S} = P^E(\bar{S}) - \left\{1 - e^{-(r-y)T} N[-d_1(\bar{S})]\right\} \frac{\bar{S}}{q_1}$$

若為不發放股利的情況，則只須令 $y = 0$ ，(6.4.4)式便可直接運用。

```

#01 '// American put
#02 Private Function BAWAmericanPutApprox(S As Double,
#03 K As Double, T As Double, r As Double, y As Double,
#04 sig As Double) As Double
#05
#06 Dim Sk As Double, n As Double, CP As Double
#07     Dim d1 As Double, Q1 As Double, a1 As Double
#08
#09     Sk = Kp(K, T, r, y, sig)
#10     n = 2 * (r - y) / sig ^ 2
#11     CP = 2 * r / (sig ^ 2 * (1 - Exp(-r * T)))
#12     d1 = (Log(Sk/K) + (r-y + sig^2/2)*T)/(sig*Sqr(T))
#13     Q1 = (-(n - 1) - Sqr((n - 1) ^ 2 + 4 * CP)) / 2
#14     a1 = -(Sk/Q1) * (1-Exp(-y* T) * NorCdf(-d1))
#15
#16     If S > Sk Then
#17         BAWAmericanPutApprox = GBSOption("p",S,K,T,
#18             sig,r,y) + a1 * (S/Sk)^Q1
#19     Else
#20         BAWAmericanPutApprox = K - S
#21     End If
#22 End Function
#23
#24
#25 '// Newton Raphson to solve critical price for a Put
#26 Private Function Kp(K As Double, T As Double,
#27 r As Double, y As Double, sig As Double) As Double
#28     Dim n As Double, m As Double
#29     Dim Su As Double, Si As Double
#30     Dim h1 As Double, A As Double
#31     Dim d1 As Double, qlu As Double, Q1 As Double
#32     Dim LHS As Double, RHS As Double
#33     Dim bi As Double, E As Double
#34
#35     '// Calculation of seed value, Si
#36     n = 2 * (r - y) / sig ^ 2
#37     m = 2 * r / sig ^ 2
#38     qlu = (-(n - 1) - Sqr((n - 1) ^ 2 + 4 * m)) / 2
#39     Su = K / (1 - 1 / qlu)
#40     h1 = ((r-y)*T - 2*sig*Sqr(T))*K/(K-Su)
#41     Si = Su + (K - Su) * Exp(h1)
#42
#43     A = 2 * r / (sig ^ 2 * (1 - Exp(-r * T)))
#44     d1 = (Log(Si/K)+(r-y + sig^2/2)*T)/(sig*Sqr(T))
#45     Q1 = (-(n - 1) - Sqr((n - 1) ^ 2 + 4 * A)) / 2
#46     LHS = K - Si

```

```

#47   RHS = GBSOption("p",Si,K,T,sig,r,y)-(1-Exp(-y*T)
#48       * NorCdf(-d1)) * Si / Q1
#49   bi = -Exp(-y*T)*NorCdf(-d1)*(1-1/Q1)-(1+Exp(-y*T)
#50       * NorPdf(-d1) / (sig * Sqr(T))) / Q1
#51   E = 0.000001
#52
#53   '// Newton Raphson to find critical price Si
#54   While Abs(LHS - RHS) / K > E
#55       Si = (K - RHS + bi * Si) / (1 + bi)
#56       d1 = (Log(Si/K)+(r-y+sig^2/2)*T)/(sig*Sqr(T))
#57       LHS = K - Si
#58       RHS = GBSOption("p",Si,K,T,sig,r,y)-(1
#59           -Exp(-y*T)*NorCdf(-d1)) * Si / Q1
#60       bi = -Exp(-y*T)*NorCdf(-d1)*(1-1/Q1)-(1+
#61           Exp(-y*T)*NorCdf(-d1)/(sig*Sqr(T)))/Q1
#62   Wend
#63
#64   Kp = Si
#65 End Function
程式 6.4.4(Book3.xls/Module1)

```

程式 6.4.4 即為 Barone-Adesi 和 Whaley(1987)的賣權演算程式碼。#01~#22 為(6.4.4)式的賣權程式碼，其中提前執行的關鍵價格 \bar{S} ，可由#25~#65 利用 Newton-Raphson 法求得。

五、間斷股利的賣權

在此情況下，美式賣權的求解相當的複雜。Geske, R.和 Johnson, H.(1984) 提供了一項近似的方法，稱之為複合式選擇權(Compound Options)。有興趣的讀者可加以參考。

